Probability Density Distribution—PROD

The program PROD (**Pro**bability Density **D**istribution) is a subroutine subprogram that calculates the probability density distribution by first normalizing the given data by its maximum value and then counting the number of sample values belonging to each of the 21 orders of magnitude between the maximum and minimum values. The mean and standard deviation of the data are calculated simultaneously.

PROD（**Pro**bability Density **Di**stribution）

【Purpose】

To compute the probability density distribution of the given data and simultaneously compute the mean and standard deviation of the data.

【Usage】

（1） How to connect

```
CALL PROD (N, X, ND, RAMP, PROP, AV,  SD)
```

Argument	Type	Parameter in calling program	Return Parameter
N	I	Total number of data X	Unchanged
X	R 1-D array (ND)	Equal interval data	Data normalized by maximum value
ND	I	Dimension size of X in calling program	Unchanged
RAMP	R 1-D array (21)	No need to input here	Representative value of relative amplitude
PROP	R 1-D array (21)	No need to input here	Probability distribution in each class (%)
AV	R	No need to input here	Average value normalized by maximum value
SD	R	No need to input here	Standard deviation normalized by maximum value

（2） Necessary subroutines and function subprograms

None

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *          PROD   1
C     SUBROUTINE FOR PROBABILITY DENSITY DISTRIBUTION                 PROD   2
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * *          PROD   3
C                                                                      PROD   4
C                                    CODED BY Y.OHSAKI                 PROD   5
C                                                                      PROD   6
      SUBROUTINE PROD(N,X,ND,RAMP,PROP,AV,SD)                          PROD   7
C                                                                      PROD   8
      DIMENSION X(ND),RAMP(21),PROP(21)                                PROD   9
      DIMENSION BOUND(21)                                              PROD  10
C                                                                      PROD  11
C     INTIALIZATION                                                    PROD  12
C                                                                      PROD  13
      DO 110 I=1,21                                                    PROD  14
      RAMP(I)=1.-REAL(I-1)/10.                                         PROD  15
      BOUND(I)=RAMP(I)-0.05                                            PROD  16
      PROP(I)=0.                                                       PROD  17
  110 CONTINUE                                                         PROD  18
      XMAX=0.                                                          PROD  19
      DO 120 M=1,N                                                     PROD  20
      XMAX=AMAX1(XMAX,ABS(X(M)))                                       PROD  21
  120 CONTINUE                                                         PROD  22
      DO 130 M=1,N                                                     PROD  23
      X(M)=X(M)/XMAX                                                   PROD  24
  130 CONTINUE                                                         PROD  25
      AV=0.                                                            PROD  26
      SD=0.                                                            PROD  27
C                                                                      PROD  28
C     PROBABILITY DENSITY                                              PROD  29
C                                                                      PROD  30
      DO 150 M=1,N                                                     PROD  31
      AV=AV+X(M)                                                       PROD  32
      SD=SD+X(M)**2                                                    PROD  33
      DO 140 I=1,21                                                    PROD  34
      IF(X(M).LT.BOUND(I)) GO TO 140                                   PROD  35
      PROP(I)=PROP(I)+1.                                               PROD  36
      GO TO 150                                                        PROD  37
  140 CONTINUE                                                         PROD  38
  150 CONTINUE                                                         PROD  39
      DO 160 I=1,21                                                    PROD  40
      PROP(I)=PROP(I)/REAL(N)*100.                                     PROD  41
  160 CONTINUE                                                         PROD  42
      AV=AV/REAL(N)                                                    PROD  43
      SD=SQRT(SD/REAL(N)-AV**2)                                        PROD  44
      RETURN                                                           PROD  45
      END                                                              PROD  46
```

【Example】

Compute the probability density distribution of El Centro seismic waves (EQ.01). The main program, for example, is as follows, and the output is shown in Table 3-1.

```
      CHARACTER NAME*50
      DIMENSION X(800),RAMP(21),PROP(21)
C
      READ(5,501) NAME,DT,NN,(X(M),M=1,NN)
      CALL PROD(NN,X,800,RAMP,PROP,AV,SD)
      WRITE(6,601) NAME,(RAMP(I),PROP(I),I=1,21)
      WRITE(6,602) AV,SD
      STOP
C
 501  FORMAT(A50,F10.0,I10/(8F10.0))
 601  FORMAT(A50//T3,'-- PROBABILLITY DENSITY DISTRIBUTION --'//
     *       T5,'REALT.AMP    PROB.DENSITY(PERCENT)'//(F13.2F16.5))
 602  FORMAT(/T5,'NORMALIZED AVERAGE VALUE',F10.3/
     *        T5,'NORMALIZED STANDARD DEV.',F10.3)
      END
```

Output：

```
EQ.01 - EL CENTRO, CALIF.  1940.5.18 NS

  -- PROBABILLITY DENSITY DISTRIBUTION --

    REALT.AMP    PROB.DENSITY(PERCENT)

         1.00              0.50
         0.90              0.75
         0.80              0.62
         0.70              0.62
         0.60              2.38
         0.50              2.62
         0.40              3.12
         0.30              5.12
         0.20              8.25
         0.10             16.38
         0.00             24.62
        -0.10             13.00
        -0.20              4.50
        -0.30              3.62
        -0.40              4.00
        -0.50              4.12
        -0.60              2.88
        -0.70              1.25
        -0.80              0.75
        -0.90              0.88
        -1.00              0.00

    NORMALIZED AVERAGE VALUE     -0.000
    NORMALIZED STANDARD DEV.      0.314
```